KANSAS GEOLOGICAL SURVEY
OPEN-FILE REPORT 89-40

Capture Zones for Simple Aquifers

by

Carl D. McElwee

KANSAS GEOLOGICAL SURVEY
1930 Constant Avenue
University of Kansas
Lawrence, KS 66047

CAPTURE ZONES FOR SIMPLE AQUIFERS

Carl D. McElwee
Kansas Geological Survey
1930 Constant Avenue
Lawrence, KS 66047

Prepared for publication in
Ground Water
December, 1989

Kansas Geological Survey
Open-File Report #89-40

ABSTRACT

The protection and cleanup of aquifers is a matter of high priority for all states and the federal government. One concept that is getting increasing attention is that of wellhead protection. Capture zones showing the area influenced by a well in a certain time are useful for aquifer protection or cleanup. If hydrodynamic dispersion is neglected, a sharp front calculation defines the capture zone. Analytical expressions for the capture zones can be derived for simple aquifers. However, the capture zone equations are transcendental and cannot be explicitly solved for the coordinates of the capture zone boundary. Fortunately, an iterative scheme allows the solution to proceed quickly and efficiently even on a modest personal computer. Three forms of the analytical solution must be used in an iterative scheme to cover the entire region of interest, after the extreme values of the x coordinate are determined by an iterative solution. The resulting solution is a discrete one and usually 100-1000 intervals along the x-axis are necessary for a smooth definition of the capture zone. The presented program is written in FORTRAN and has been used in a variety of computing environments. No graphics capability is included with the program; it is assumed the user has access to a commercial package. The superposition of capture zones for multiple wells is expected to be satisfactory if the spacing is not too close. Since this program deals with simple aquifers, the results will rarely be the final word in a real application. However, the program is very useful as a first phase in developing wellhead protection or aquifer cleanup schemes.

INTRODUCTION

The protection and cleanup of aquifers is a matter of high priority for all states and the federal government, as evidenced by the large number of laws and regulations that have been created in recent years. One concept that is getting increasing attention is that of wellhead protection, where certain potentially polluting activities are banned or highly regulated within an area that would affect a well within a certain time period. In terms of aquifer cleanup, one would like to know what area will be pumped out of an aquifer by a discharge well in a certain time period. These areas are commonly referred to in the literature as capture zones. Since real world aquifers are very complex, exhibiting heterogeneity and other complicating factors, the calculation of realistic capture zones is not easy. Possible techniques range from simple analytical methods to complex numerical procedures. If hydrodynamic dispersion is neglected, a sharp front calculation is used to define the capture zone.

The calculation of sharp front movement for wells in infinite aquifers dates at least to Muskat (1937). More recently Bear and Jacobs (1965) have investigated the movement of water bodies injected into isotropic homogeneous aquifers with uniform regional flow by analytical methods. Most groundwater texts present a steady-state analytical solution for the groundwater divide in an isotropic homogeneous aquifer with one pumping well located in a uniform regional flow field (see for example Todd, 1980); this corresponds to an infinite-time capture zone. Javandel et al. (1984) present semi-analytical methods for calculating pathlines and time related capture zones for multiple wells in simple aquifers (isotropic, homogeneous, uniform thickness, uniform regional flow, and steady state). However, their computer program is rather complex. Javandel and Tsang (1986) propose infinite-time capture zone curves as a tool for aquifer cleanup; again, they use analytical methods for simple aquifers. A few authors have utilized numerical methods to calculate time-related capture zones in the presence of aquifer heterogeneity (Kinzelback, 1986; and Shafer, 1987).

The purpose of the present paper is to present a program for calculating time-related capture zones in simple aquifers. The program is short and efficient and adaptable to a range of computing environments from personal computers to main frames. Since the program assumes simple aquifer conditions (isotropic, homogeneous, uniform thickness, uniform regional flow, and steady state), it should be used with care in a real-world situation. However, the program should be a useful initial planning tool for aquifer protection or cleanup.

BASIC EQUATIONS

The basic equations that are used to describe the capture-zone curves are taken from Bear and Jacobs (1965). As mentioned

in the introduction, this formulation assumes an aquifer with a constant regional hydraulic conductivity (K). A regional flow direction and magnitude (q_o) is also assumed constant and given by the Darcy equation.

$$q_o = -K \frac{\partial h}{\partial s} \tag{1}$$

h is the regional hydraulic head (without the pumping well) and s is the direction of the head gradient perpendicular to lines of constant head. The aquifer is assumed to be of constant thickness (B) and constant porosity (n). At this point, a well pumping at a rate Q is superimposed upon the regional system and a new steady-state head configuration is established. The object is to calculate the area of the aquifer that will contribute water to the well during a specified time period, or alternatively, to calculate the area affected by injection for a given time interval, after the well is in steady state with the regional system. The curves surrounding these affected areas are loosely called capture curves for a given time period. It is convenient to define three dimensionless parameters.

$$\bar{x} = \frac{2\pi q_o B}{Q} x \tag{2}$$

$$\bar{y} = \frac{2\pi q_o B}{Q} y \tag{3}$$

$$\bar{t} = \frac{2\pi q_o^2 B}{Q} t \tag{4}$$

x, y, and t are the space and time coordinates in the real world; whereas, $\bar{x}$, $\bar{y}$, and $\bar{t}$ are their dimensionless counterparts. Using these dimensionless variables, Bear and Jacobs (1965) show that the capture curves are given by the following equation.

$$\exp(\bar{x} - \bar{t}) = \cos \bar{y} + \frac{\bar{x}}{\bar{y}} \sin \bar{y} \tag{5}$$

Unfortunately, equation 5 is a transcendental equation which cannot be solved explicitly for either $\bar{x}$ or $\bar{y}$.

There are two cases where equation 5 can be simplified somewhat. When $t \to \infty$ equation 5 reduces to

$$\bar{x} = -\frac{\bar{y}}{\tan \bar{y}} , \tag{6}$$

which is the familiar form for the groundwater divide (Todd, 1980). From equation 6, it is seen that as $\bar{x} \rightarrow \infty$ the limiting value of $\bar{y}$ is $\pm\pi$. The stagnation point occurs at $\bar{y} = 0$ and $\bar{x} = -1$; this result can be obtained from equation 6 by taking the limit as $\bar{y} \rightarrow 0$. Another useful simplification of equation 5 results when $\bar{y} = 0$; this corresponds to the two points where the capture curve crosses the $\bar{x}$ axis. Taking the limit as $\bar{y} \rightarrow 0$ of equation 5 gives the extreme values $\bar{x}_e$,

$$\exp(\bar{x}_e - \bar{t}) = 1 + \bar{x}_e. \tag{7}$$

A slight rearrangement of equation 7 gives the form

$$\bar{t} = \bar{x}_e - \ln(1 + \bar{x}_e). \tag{8}$$

Equations 5 and 8 will form the basis for calculating capture curves at a given normalized time $\bar{t}$. Unfortunately, both are transcendental equations, so iterative techniques will be used to obtain their solution. The resulting curves will enclose the area of the aquifer containing water either injected or discharged by the well up to time $\bar{t}$. The curves represent sharp fronts because hydrodynamic dispersion has been neglected.

ITERATIVE SOLUTION

The capture curves given by equation 5 are symmetric about the $\bar{x}$ axis which is assumed to be parallel to the regional hydraulic gradient with its origin at the well. From discussion in the previous section, the limits on the coordinates are $-1 \leq \bar{x} \leq \infty$ and $-\pi \leq \bar{y} \leq \pi$. The solution that we shall obtain is a numerical one at discrete values of $\bar{x}$. The approach that we shall take involves solving equation 8 for the extreme values of $\bar{x}$ at a certain $\bar{t}$. The region bounded by these two extremes will be discretized to give a certain number of discrete values of $\bar{x}$ (usually between 100 and 1000). Let $\bar{x}_i$ represent one of these values. At that point with $\bar{t}$ and $\bar{x}$ known, equation 5 will be solved to obtain $\bar{y}$. Unfortunately, no single form of equation 5 seems to work well for the full range of coordinates.

The first order of business is to find the extreme values of $\bar{x}$ by solving equation 8. Rearranging equation 8 slightly allows an iterative solution scheme to be developed (one-point method, Atkinson, 1989).

$$\bar{x}_e^{(m+1)} = \bar{t} + \ln[1 + \bar{x}_e^{(m)}] \tag{9}$$

The m in equation 9 is an iteration index. An initial guess for $\bar{x}_e$ must be known, but $\bar{x}_e = 0$ always seems to work well.

Iteration continues on equation 9 until convergence occurs. If the initial guess for $\bar{x}_e$ is zero and $\bar{t}$ is positive, it is clear that equation 9 will converge on a positive value. If $\bar{t}$ is small then $\bar{x}_e$ will also be small and the log term of equation 8 can be written as a series expansion to yield

$$\bar{t} = \bar{x}_e - \bar{x}_e + \frac{\bar{x}_e^2}{2} - \frac{\bar{x}_e^3}{3} + \frac{\bar{x}_e^4}{4} - \cdots \tag{10}$$

Solving for the lowest power of $\bar{x}_e$ gives

$$\bar{x}_e^{(m+1)} = \sqrt{2}\left[\bar{t} + \frac{\bar{x}_e^{(m)3}}{3} - \frac{\bar{x}_e^{(m)4}}{4} + \cdots\right]^{1/2} \tag{11}$$

Iterating equation 11 works well for small values of $\bar{t}$ and $\bar{x}_e$.

Equations 9 and 11 work well for the positive value of the $\bar{x}$ extremes; however, a slightly different version is needed to find the negative extreme value. Rearranging equation 7 slightly gives the following iterative solution.

$$\bar{x}_e^{(m+1)} = \exp(\bar{x}_e^{(m)} - \bar{t}) - 1 \tag{12}$$

Clearly, if $\bar{t} \to \infty$ equation 12 gives an extreme value of -1. If the initial guess for $\bar{x}_e$ is zero and $\bar{t}$ is positive the result for the first iteration will be negative. Experience has shown that equation 12 converges rapidly on the negative value of $\bar{x}_e$.

Now that the extreme values of $\bar{x}$ are known for a particular $\bar{t}$, we can pick a discrete value $\bar{x}_i$ located between these two extremes. The only unknown in equation 5 is now $\bar{y}$ and an iterative solution can be set up. The most obvious iterative form is obtained from equation 5 by multilpying by $\bar{y}$ and $\exp(\bar{t} - \bar{x})$ to obtain

$$\bar{y}_i^{(m+1)} = \exp(\bar{t} - \bar{x}_i) \cdot \left[\bar{y}_i^{(m)} \cos \bar{y}_i^{(m)} + \bar{x}_i \sin \bar{y}_i^{(m)}\right] . \tag{13}$$

However, numerical experiments show that equation 13 does not converge for all values of $\bar{x}$, $\bar{y}$, and $\bar{t}$. As long as $|\bar{y}_i| \leq \pi/2$

and $\bar{x} \geq 1$ equation 13 works well.

An alternate form of equation 5 can be obtained by solving for cos $\bar{y}$ and then taking the inverse cosine function.

$$\bar{y}_i^{(m+1)} = \cos^{-1}\left[\exp(\bar{x}_i - \bar{t}) - \frac{\bar{x}_i}{\bar{y}_i^{(m)}}\sin \bar{y}_i^{(m)}\right] \tag{14}$$

Numerical experiments show that this form works well for all values of $\bar{x}$ and $\bar{y}$ as long as $\bar{t} \leq 1$. For $\bar{t} \geq 1$ equation 14 can be used only for $\bar{x} \leq 1$.

The final form of equation 5 needed to fill in all remaining values of $\bar{x}$, $\bar{y}$, and $\bar{t}$ is given by rearranging and solving for the tangent of $\bar{y}_i$.

$$\tan \bar{y}_i = \left(\frac{\bar{y}_i}{\bar{x}_i}\right) \cdot \left(\frac{\exp(\bar{x}_i - \bar{t})}{\cos \bar{y}_i} - 1\right) \tag{15}$$

Using the trigonometry identity $\tan(-\theta) = \tan(\pi - \theta)$ allows us to rewrite equation 15 in iterative notation.

$$\bar{y}_i^{(m+1)} = \pi - \tan^{-1}\left[\left(\frac{\bar{y}_i^{(m)}}{\bar{x}_i}\right) \cdot \left(1 - \frac{\exp(\bar{x}_i - \bar{t})}{\cos \bar{y}_i^{(m)}}\right)\right] \tag{16}$$

Numerical experiments show that this equation works well for $\bar{t} > 1$ and $\bar{x} > 1$ if $|\bar{y}_i| > \pi/2$. Clearly, equation 16 has a problem at $\bar{y} = \pi/2$ since the cosine function is zero. Therefore, special provision must be made to prevent equation 16 from being used too near the region where $\bar{y} = \pi/2$.

The iterative equations 13, 14, and 16 for $\bar{y}$ require an initial guess for the $m = 0$ iteration. That question was avoided in the above paragraphs where the equations were developed. However, in practice this presents no problem. Using the extreme values of $\bar{x}$, a discrete set of $\bar{x}_i$'s are calculated by dividing the region between the extremes into an integral number of steps (usually between 100 and 1000). Solution then proceeds sequentially from the negative $\bar{x}$ extreme to the positive $\bar{x}$ extreme. At each of the extreme values of $\bar{x}$ we know that $\bar{y} = 0$.

Therefore, as we step through the solution we will always know the value of $\bar{y}$ at the previous $\bar{x}$ value and we can use this as the initial guess for $\bar{y}$ at the current value of $\bar{x}$. If at least 100 steps in $\bar{x}$ are used, the value of $\bar{y}$ does not change dramatically in one step and the above procedure is very efficient. As the solutionproceeds, the appropriate equation (13, 14, or 16) is selected depending on the values of $\bar{t}$, $\bar{x}$, and the current value of $\bar{y}$.

COMPUTER PROGRAM

A computer program to calculate capture curves based on the material presented here has been written in FORTRAN and is included in the appendix. We commonly run it on an IBM AT compatible computer; however, it can be easily adapted to a wide variety of computer environments, usually only the OPEN command for the output file will need modification. The program consists of three parts, the main program and two subroutines. The three parts are briefly described in the following paragraphs.

The main program interacts with the user via the screen to read in some necessay information. The first request is for the user to supply a value of $\bar{t}$. Next, the user must specify the number of intervals between the two $\bar{x}$ extremes (N). If N is specified, there will be N+1 values of $\bar{x}_i$ calculated, including the two extreme values. The $\bar{x}$ and $\bar{y}$ values for a given $\bar{t}$ will always be written to a computer file; however, the user may have the values echoed to the screen if one responds with YES at the prompt. The data file to receive the calculated values must be specified next. As currently written, the prompt asks for the path name in the usual IBM format; that can easily be changed. The output file should be given a descriptive name to identify it, since only $\bar{x}$ and $\bar{y}$ values are written by the program. After opening the file the program calls the subroutine CAPCUR to calculate the values of $\bar{x}$ and $\bar{y}$ for a given $\bar{t}$. CAPCUR only calculates the positive values of $\bar{y}$, so the main program fills in the negative values for all the $\bar{x}$'s. Since the capture curves are symmetric about the $\bar{x}$-axis, that does not involve more calculations. After closing the output file, the main program asks for a new value of $\bar{t}$. If $\bar{t}$ is less than zero the program terminates execution.

The subroutine CAPCUR has five arguments: T, N, X, Y, and ECHO. T, N, and ECHO are user supplied and are passed from the main program. T is the value of $\bar{t}$ and N is the number of intervals to be used in calculating the $\bar{x}_i$. If ECHO is set to YES, the calculated values of $\bar{x}$ and $\bar{y}$ will be written to the screen. Currently, the arrays for X and Y are dimensioned to a maximum of 1001, which means that N can not be larger than 1000. That can easily be changed, however, we have found 1000 to be sufficient even for large values of $\bar{t}$ up to about 1600. For larger values of $\bar{t}$ the arrays will need to be dimensioned larger. A good rule is that the dimension of the arrays should be at

least .6 x max $\bar{t}$. One should not give N less than 100, because we have found that the capture curves may not have an adequate sample to be represented smoothly. CAPCUR calls ENDPTS which calculates the extreme values of $\bar{x}$ from equations 9, 11, and 12. (Of course $\bar{y}$ = 0 at each of these extreme values of $\bar{x}$.) The interval (DX) between various $\bar{x}_i$ is calculated by dividing the region between the extremes by N. The DO loop ending on statement 150 calculates the current $\bar{x}_i$ by adding DX and then calculates the corresponding $\bar{y}_i$ by using the appropriate iterative equation (13, 14, or 16). The program requires the first $\bar{x}_i$ in the loop (i = 2) to be less than 1.0 so that equation 14 can be used for the first calculation. If this is not the case, the program states that N is too small and stops. The initial guess for $\bar{y}_{i+1}$ is simply the value of $\bar{y}_i$ at the preceding point. The appropriate iterative equation is selected by testing $\bar{t}$, $\bar{x}$, and $\bar{y}$. If the iterative procedure does not converge in 2000 iterations a message will be printed and the program stops. Failure to converge will rarely occur, but could happen if N is chosen too small. Once all $\bar{y}_i$ are calculated for all $\bar{x}_i$, the subroutine returns control to the main program.

The subroutine ENDPTS calculates the extreme values of $\bar{x}$ for a given $\bar{t}$. It is a straight forward application of the iterative equations 9, 11, or 12. An initial guess of $\bar{x}$ = 0 is used. Equation 9 or 11 is used for the positive extreme value of $\bar{x}$, depending on the value of $\bar{t}$. For the negative solution, equation 12 is used. These extreme values are then returned to the calling program.

RESULTS AND APPLICATION

The results of using the algorithms discussed here are shown in figure 1 for $\bar{t}$ values of 1, 3, 5, and ∞. The $\bar{t} = \infty$ curve corresponds to the normal groundwater divide. Equations 13, 14, and 16 can only be applied in certain regions of $\bar{t}$, $\bar{x}$, and $\bar{y}$ as discussed earlier. These various regions are shown on figure 1, each with a different background pattern. Figure 1 was produced with a commercially available graphics package directly from the output file of the program in the appendix. No graphics capability is included in the program, it is assumed that the user has access to a similar package.

In a real world application, one will not be dealing with the dimensionless quantities $\bar{t}$, $\bar{x}$, and $\bar{y}$ but with actual time and distances. However, equations 2, 3, and 4 provide the necessary conversions, so it is easy to adapt the current program to a specific application. To begin, one must know the average value of hydraulic conductivity and the regional hydraulic gradient vector (direction and magnitude). These quantities are used in equation 1 to calculate the specific discharge or Darcy velocity (q_o). Knowing the average regional thickness of the aquifer (B) and the discharge (or injection) rate of the well (Q), one can calculate $\bar{t}$ from equation 4 for the actual time of

interest. The program can then be run to calculate the $\bar{x}_i$ and $\bar{y}_i$ of the capture curve of interest. These values of $\bar{x}$ and $\bar{y}$ can be used with equations 2 and 3 to solve for the real world coordinates x and y, which can then be plotted on an appropriate map base. Currently, the program assumes that the x axis is parallel to the regional hydraulic gradient and that the well is located at the origin of the coordinate system. If that is not true, an appropriate rotation and translation of coordinates will be needed before plotting on the desired map base.

DISCUSSION

Strictly speaking, the program presented here only deals with one well in a uniform, homogeneous, isotropic aquifer with uniform, steady, regional flow. In practice these conditions are rarely satisfied. However, the type of analysis presented here can be very useful as a first phase in developing wellhead protection or aquifer cleanup schemes (Javandel and Tsang, 1986). If conservative aquifer parameters are used, the analysis presented here should outline a maximum capture zone. The program presented here only deals with one well; however, the approximate result for several wells can be obtained by applying the program once for each well and superimposing the results. As long as the capture zones do not overlap, the approximate result should be very good. As the well spacing gets smaller and the capture zones overlap the approximate results will deviate more from the correct solution; as long as the well spacing is greater than or equal to $Q/\pi q_o B$ the results are expected to be acceptable (see Javandel and Tsang for details of superimposing multiple wells). For the final analysis, if heterogeneity and nonuniform flow are very important a more complex program such as that presented by Shafer (1987) should be used

The program presented here is useful for planning wellhead protection and aquifer cleanup schemes. However, the user must always be mindful of its limitations. The presented program is simple and can be embedded in many computing environments, including personal computers, work stations, and main frames. We have used the program on a work station interfaced with a geographical information system (GIS) to plot capture zones for several wells in Kansas (Woods et al., 1987; Merchant et al., 1987; and Merchant et al., 1988). The program is presented here in the hope that it will be useful to others.

REFERENCES

Atkinson, K.E. 1989. An Introduction to Numerical Analysis. John Wiley, N.Y. 693 pp.

Bear, J. and M. Jacobs. 1965. On the movement of water bodies injected into aquifers. Journal of Hydrology. v. 3, pp. 37-57.

Javandel, I., C. Doughty, and C.F. Tsang. 1984. Groundwater Transport: Handbook of Mathematical Models. Water Resources Monograph Series 10. American Geophysical Union, Washington, D.C. 228 pp.

Javandel, I. and C.F. Tsang. 1986. Capture-zone type curves: A tool for aquifer cleanup. Ground Water. v. 24, no. 5, pp. 616-625.

Kinzelbach, W. 1986. Groundwater Modeling. Elsevier, Amsterdam. 333 pp.

Merchant, J.W., D.O. Whittemore, J.L. Whistler, C.D. McElwee, and J.J. Woods. 1988. Groundwater pollution hazard assessment: A GIS approach. Proceedings of the International Geographic Information Systems (IGIS) Symposium. Association of American Geographers, Washington, D.C. pp. _____.

Muskat, M. 1937. The Flow of Homogeneous Fluids Through Porous Media. McGraw Hill, NY. 763 pp.

Shafer, J. M. 1987. Reverse pathline calculation of time-related capture zones in non uniform flow. Ground Water. v. 25, no. 3, pp. 283-289.

Todd, D. K. 1980. Groundwater Hydrology. John Wiley and Sons, NY. 535 pp.

Whittemore, D.O., J.W. Merchant, J. Whistler, C.D. McElwee, and J.J. Woods. 1987. Groundwater protection planning using the ERDAS geographic information system: Automation of DRASTIC and time-related capture zones. Proceedings of the FOCUS Conference on Midwestern Groundwater Issues. National Water Well Association, Dublin, Ohio. pp. 359-374.

Woods, J.J., C.D. McElwee and D.O. Whittemore. 1987. Computation of time-related capture zones of wells for use with the ERDAS geographic information system. Kansas Geological Survey, Lawrence, KS. Open-File Report no. 87-14, 59 pp.

APPENDIX

```
      CHARACTER*20 NAME
      CHARACTER*3 ECHO
      DIMENSION X(1001), Y(1001)
   10 WRITE (*,101)
  101 FORMAT (' READ IN A T VALUE')
      READ (*,*) T
      IF (T .LT. 0.0) STOP
      WRITE (*,106)
  106 FORMAT (' READ IN THE NUMBER OF CALCULATION POINTS, N.',/
     + ,' N SHOULD BE BETWEEN 100 AND 1000.')
      READ (*,*) N
      WRITE (*,107)
  107 FORMAT (' DO YOU WANT TO ECHO THE RESULTS TO THE SCREEN?')
      READ (*,203) ECHO
  203 FORMAT (A3)
      WRITE (*,108)
  108 FORMAT (' READ IN THE FILE NAME TO SAVE DATA IN.'
     + ,' MUST BE OF THE FORM [C:\DIRECT\NAME.DAT].')
      READ (*,202) NAME
  202 FORMAT (A20)
      OPEN (99,FILE=NAME,STATUS='UNKNOWN')
      CALL CAPCUR(T,N,X,Y,ECHO)
      DO 300 I = 1, N+1
      WRITE (99,201) X(I), Y(I)
  201 FORMAT (2E20.8)
  300 CONTINUE
      DO 400 I = N, 1, -1
      WRITE (99,201) X(I), -Y(I)
  400 CONTINUE
      CLOSE (99)
      GO TO 10
      END


      SUBROUTINE CAPCUR(T,N,X,Y,ECHO)
      CHARACTER*3 ECHO
      DIMENSION X(1001), Y(1001)
      PID2 = 1.5707963
      PID2P = PID2 + .01
      CALL ENDPTS(T,XMIN,XMAX)
      X(1) = XMIN
      Y(1) = 0.0
      IF (ECHO .EQ. 'YES')  WRITE (*,103) XMIN, 0.0, 0
  103 FORMAT (' FOR X =',E20.8,' THE Y = ',E20.8
     + ,' NEQ = ',I2)
      NEQ = 1
      DX = (XMAX - XMIN)/N
      YN = 0.1
```

```
    DO 150 I = 2, N
    XI = XMIN + DX*(I-1)
    IF (XI .GT. 1.0 .AND. I .EQ. 2 ) THEN
    WRITE (*,101)
101 FORMAT( '  N HAS BEEN CHOSEN TOO SMALL.')
    STOP
    ENDIF
    II = 0
    IF ( T .LE. 1.0) GO TO 19
    IF (XI .LE. 1.0) GO TO 19
    IF (NEQ .EQ. 3) GO TO 40
    IF ((XI .GT. 1.0) .AND. (YN .GT. PID2)) GO TO 30
    IF ((XI .GT. 1.0) .AND. (YN .LT. PID2)) GO TO 40
 19 NEQ = 1
 20 YI = YN
    II = II + 1
    IF (II .GT. 2000) GO TO 200
    YN = ACOS(EXP(XI-T)-(XI/YI)*SIN(YI))
    YN = ABS(YN)
    IF (ABS((YN-YI)/(YN+.00001)) .GT. .00001) GO TO 20
    IF (ECHO .EQ. 'YES')  WRITE (*,103) XI, YN, NEQ
    GO TO 130
 30 NEQ = 2
 35 YI = YN
    II = II + 1
    IF (II .GT. 2000) GO TO 200
    YN = 3.14159-ATAN((YI/XI)*(1.0-EXP(XI-T)/COS(YI)))
    YN = ABS(YN)
    IF (ABS((YN-YI)/(YN+.00001)) .GT. .00001) GO TO 35
    IF (ECHO .EQ. 'YES')  WRITE (*,103) XI, YN, NEQ
    IF ( YN .LT. PID2P) NEQ = 3
    GO TO 130
 40 NEQ =3
 45 YI = YN
    II = II + 1
    IF (II .GT. 2000) GO TO 200
    YN = EXP(T-XI)*(YI*COS(YI) + XI*SIN(YI))
    YN = ABS(YN)
    IF (ABS((YN-YI)/(YN+.00001)) .GT. .00001) GO TO 45
    IF (ECHO .EQ. 'YES')  WRITE (*,103) XI, YN, NEQ
    GO TO 130
200 WRITE (*,105)
105 FORMAT (' DID NOT CONVERGE IN 2000 ITERATIONS. STOP')
    STOP
130 X(I) = XI
    Y(I) = YN
150 CONTINUE
    X(N+1) = XMAX
    Y(N+1) = 0.0
    IF (ECHO .EQ. 'YES')  WRITE (*,103) XMAX, 0.0, 0
    RETURN
    END
```

```
      SUBROUTINE ENDPTS(T, XN1, XN2)
      XN = 0.0
C POSITIVE X SOLUTION
      IF (T .LT. .10) GO TO 90
   30 X = XN
      XN = T + LOG(1.0 +X)
      IF (ABS((XN - X)/XN) .GT. .00001) GO TO 30
      XN2 = XN
      GO TO 140
   90 CONTINUE
   95 X = XN
      XN=SQRT(2.*(T+X**3/3.-X**4/4.+X**5/5.-X**6/6.
     * +X**7/7.-X**8/8.+X**9/9.-X**10/10.))
      IF (ABS((XN-X)/(XN+.00001)) .GT. .00001) GO TO 95
      XN2 = XN
C NEGATIVE X SOLUTION
  140 XN = 0.0
  160 X = XN
      XN = EXP(X-T) -1.
      IF (ABS((XN-X)/(XN+.00001)) .GT. .00001) GO TO 160
      XN1 = XN
      RETURN
      END
```

Figure 1. Capture zones and regions of application.